VIRTUAL

SAS® GLOBAL FORUM 2021

#SASGF

Bayesian Analysis of GLMMs Using PROC BGLIMM

Walt Stroup, University of Nebraska-Lincoln

Bio:
Emeritus Professor of Statistics, University of Nebraska-Lincoln.
Co-author *SAS for Mixed Models* (1996, 2006, 2018).
Author *Generalized Linear Mixed Models* (2013).
Teaching & Research specializations:
statistical modeling, design of experiments,
collaboration with allied disciplines ("consumers of statistical methods").
ASA Fellow. Founding chair: UNL Department of Statistics.

Outline

I. GLMM Basics

A. Defining Elements of a GLMM

B. Overview of Bayesian Estimation and Inference for GLMMs

II. Three Examples

A. Multi-Clinic Binomial (Beitler-Landis Data from SAS MM 2018 ed, Ch 11)

i. Preliminary PROC BGLIMM run

ii. Diagnostics

iii. Needed options

iv. Revised PROC BGLIMM run, useable analysis, interpretation

v. Post-processing steps (equivalent of ILINK in PROC GLIMMIX)

B. Multi-level (a.k.a. split-plot or hierarchical) with count data (SAS MM Ch 13)

C. Repeated measures (a.k.a. longitudinal) (SAS MM, Ch 8)

Generalized Linear Mixed Model (GLMM) Setting

Response Variable Distribution - Model Effect Combinations

Classic Statistical Model Format:
response = systematic + random/residual

response variable		systematic/explanatory $X\beta$ Categorical e.g. $\mu + \tau_i$	systematic/explanatory $X\beta$ continuous e.g. $\beta_0 + \beta_1 X$	random/residual Zb model effects	random/residual Residual error structure (e.g. serial/spatial)
Gaussian (normal)					"R-side"
Non-Gaussian	**discrete proportion** binomial multinomial				"G-side"
Non-Gaussian	**continuous proportion** beta				
Non-Gaussian	**count** Poisson negative Binomial				
Non-Gaussian	time to event				
Non-Gaussian	etc....				

Defining Elements of a GLMM

- $\boldsymbol{\eta} = g(\boldsymbol{\mu}) = \boldsymbol{X\beta} + \boldsymbol{Zb}$ (link function = linear predictor)
- $\boldsymbol{b} \sim N(0, \boldsymbol{G})$
- $\boldsymbol{y}|\boldsymbol{b} \sim \mathcal{D}(\boldsymbol{\mu}, \boldsymbol{\Sigma})$
- $\boldsymbol{\Sigma} = \boldsymbol{V}_{\mu}^{1/2} \boldsymbol{R} \boldsymbol{V}_{\mu}^{1/2}; \boldsymbol{V}_{\mu}^{1/2} = diag[v(\mu)]$
- $\boldsymbol{R}$ is scale matrix, e.g. $\boldsymbol{R} = \boldsymbol{I\phi}$ or repeated measures covariance
- log likelihood $(y|b) \propto exp\left[\frac{y\theta - b(\theta)}{\phi}\right]$
- $\mu = \frac{\partial b(\theta)}{\partial \theta}; v(\mu) = \frac{\partial^2 b(\theta)}{\partial \theta^2}; Var(y|b) = \phi v(\mu)$
- canonical parameter $\theta(\mu)$ ➔ $\theta_{\eta} = \theta[g^{-1}(\boldsymbol{X\beta} + \boldsymbol{Zb})]$

GLMM Estimation and Inference

- Frequentist approach: maximum likelihood
- Log likelihood:
  - $\log[f(y;\beta)]$
  - $f(y;\beta) = \int_b f(y;\beta|b)f(b)db$
- Log likelihood generally intractable
  - Linearization – pseudo-likelihood in PROC GLIMMIX
  - Integral approximation – quadrature or Laplace (GLIMMIX, NLMIXED)
- Obtain estimates and standard errors
  - Pseudo-likelihood allows REML (or REML-like) covariance estimates
  - Use estimates & SE to compute test statistics or confidence intervals

Bayesian Estimation and Inference

- Posterior distribution (heuristic overview)
  - $f(\beta,\sigma|y,b) = \frac{f(y,b|\beta,\sigma)f(\beta,\sigma)}{\int f(y,b|\beta,\sigma)f(\beta,\sigma)d\beta d\sigma}$
  - σ is vector of scale & covariance parameters to be estimated
  - $f(y,b|\beta,\sigma)$ defined by GLMM distributions $f(y|b)$, $f(b)$
  - Generally intractable: approximated by simulation
- Steps
  - specify GLMM [defines $f(y,b|\beta,\sigma)$] and priors [define $f(\beta,\sigma)$]
  - tuning
  - burn-in
  - sample posterior distribution $\Rightarrow$ mean, median, quantiles, credible interval

Why Bayesian Methods & Not PROC GLIMMIX?

- Academic Journals
  - some prefer Bayesian credible intervals to classical significance tests
  - some require Bayesian analysis
- More generally applicable reasons
  - Frequentist MLE and Bayesian estimates with “non informative prior” are essentially equivalent
    - $\Rightarrow$ frequentist MLE implicitly assumes ignorance until data analyzed
    - (arguably) compromises efficiency and accuracy
  - In reality, studies do not begin in complete ignorance
  - In many cases, a study has been preceded by many similar studies
  - Moderately informative priors allow us to “use what we know”

Example 1: Multi-clinic, Binomial Data

- Example 11.5, *SAS for Mixed Models* (2018)
- Beitler and Landis (*Biometrics*, 1985)
- Eight clinics; sample representative of target population
- Two treatments: CNTL and DRUG
- Two groups at each clinic, n_{ij} subjects assigned to i^{th}treatment at j^{th} clinic
- Response variable: y_{ij} = number of patients with favorable outcome out of n_{ij} in group

Blocked Designs

- **Types**
  - **Multi-location, multi-center, multi-clinic**
  - **Matched pairs**
  - **Before and after on same subject**
  - **Field plots**
  - **etc.**
- **Visualization**

Treatment design	
Trt 1	Trt 2

"Experiment" (study) design		
Clinic	Group	
1		
2		
3		
4		
....		
8		

Full design		
Clinic	Group	
1	trt 1	trt 2
2	trt 2	trt 1
3	trt 2	trt 1
4	trt 1	trt 2
....		
8	trt 2	trt 1

Repurposed ANOVA

Experiment		Treatment		Combined	
Source	d.f.	Source	d.f.	Source	d.f.
clinic	7			clinic	7
		trt	1	trt	1
group(clinic)	8	"parallels" (Yates, Fisher, 1935)	14	group(clinic) | trt a.k.a. "residual" a.k.a "clinic x trt"	8-1=7
Total	15	Total	15	Total	15



Resulting Logit-normal GLMM

combined

Source	d.f.
clinic	7
treatment	1
Group(clinic) | trt	
clinic x trt	
"residual"	7
total	**15**

$\Rightarrow c_j$ i.i.d. $N\left(0,\sigma_c^2\right); ct_{ij}$ i.i.d. $N\left(0,\sigma_{ct}^2\right)$

Linear predictor: $\eta+\tau_i+c_j+\left(ct\right)_{ij}$

Link: $\eta_{ij}=\log\left(\pi_{ij}\Big/1-\pi_{ij}\right)$

$y_{ij}\mid c_j, ct_{ij} \sim \text{ind Binomial}\left(N_{ij},\pi_{ij}\right)$

$\hat{\pi}_{ij}=1\Big/\left\{1+\exp\left[-\left(\hat{\eta}+\hat{\tau}_i+\hat{c}_j+ct_{ij}\right)\right]\right\}$

FYI: Beta-binomial Mixed Model

combined

Source	d.f.
clinic	**7**
treatment	1
group (clinic) | trt "residual"	7
total	**15**

c_j i.i.d. $N\left(0, \sigma_c^2\right)$

Linear predictor: $\eta + \tau_i + c_j$

Link: $\eta_{ij} = \log\left(\mu_{ij} / 1 - \mu_{ij}\right)$

$y_{ij} \mid c_j, p_{ij} \sim \text{ind Binomial}\left(N_{ij}, p_{ij}\right)$

$p_{ij} \sim \text{beta}\left(\mu_{ij}, \varphi\right)$ or beta $\left(\alpha_{ij}, \beta_{ij}\right)$

$\hat{\mu}_{ij} = 1 / \left\{1 + \exp\left[-\left(\hat{\tau}_i + \hat{c}_j\right)\right]\right\}$

$\alpha_{ij} = \mu_{ij}\varphi;\ \beta_{ij} = \left(1 - \mu_{ij}\right)\varphi$

No ct_{ij} - "residual" info used to estimate φ (beta scale parameter)

PROC BGLIMM requires Gaussian random effects; p_{ij} not Gaussian

Beta-binomial model requires PROC MCMC (another time...)

Re-purposed ANOVA and Sensible Model

sensible model ➔ one-to-one ANOVA effect – model parameter match

combined		model				
Source	d.f.	LMM	naive GLMM	Logit-normal GLMM w/unit	Beta Binomial GLMM	Beta Binomial GLMM w/unit
clinic	7	c_j	c_j	c_j	c_j	c_j
treatment	1	τ_i	τ_i	τ_i	τ_i	τ_i
group | trt clinic x trt "residual"	7	e_{ij} or σ^2	➔ overdispersion likely	ct_{ij}	ϕ	ϕ, ct_{ij} **confounded**
total	**15**					

Re-purposed ANOVA and Sensible Model

sensible model ➔ one-to-one ANOVA effect – model parameter match

combined		model				
Source	d.f.	LMM	naive GLMM	Logit-normal GLMM w/unit	Beta Binomial GLMM	Beta Binomial GLMM w/unit
clinic	7	c_j	c_j	c_j	c_j	c_j
treatment	1	τ_i	τ_i	τ_i	τ_i	τ_i
group | trt clinic x trt "residual"	7	e_{ij} or σ^2	➔ overdispersion likely	ct_{ij}	ϕ	ϕ, ct_{ij} confounded
total	**15**					

PROC BGLIMM statements for Example 1

PROC GLIMMIX

```
/* as shown in SAS MM 3 */
proc glimmix data=clinics;
 class clinic trt;
 model fav/Nij=trt;
 random intercept trt
    / subject=clinic;
 lsmeans trt / ilink diff
    oddsratio cl;
run;
```

PROC BGLIMM

```
proc bglimm data=clinics seed=81152097
   plots=(trace autocorr density)
   diagnostics=all outpost=cout
   statistics(percentage=(2.5 50 97.5)) dic;
 class clinic trt;
 model fav/Nij=trt;
 random  intercept trt / subject=clinic;
 estimate 'cntl' intercept 1 trt 1 0;
 estimate 'drug' intercept 1 trt 0 1;
 estimate 'log odds ratio' trt 1 -1;
 ods output estimates=model_scale;
run;
```

PROC BGLIMM Diagnostics

- Trace (a.k.a. “Caterpillar”) and autocorrelation plots
- Geweke
  - compares estimates from early and late iterations
  - must be “acceptably similar”
- Heidelberger-Welch
  - stationarity test
  - half-width test – does sample size provide acceptable parameter estimate accuracy?
- Raftery-Lewis
  - addresses accuracy of posterior distribution’s percentile estimates
  - Dependence factor measures posterior sample independence ($\cong 1$ is ideal)
- Effective Sample Size
  - autocorrelation in sampling algorithm limits efficiency
  - number of samples taken versus adjusted number after accounting for autocorrelation

Plots from BGLIMM Run with Defaults

What Default Run Produces



What you should see



What you should see

Trace Plot: "fuzzy caterpillar"

Autocorr: quickly → 0 as lag increases

Worrisome Diagnostics from Default Run, 1 of 3

Heidelberger-Welch Diagnostics								
	Stationarity Test				Half-Width Test			
Parameter	Cramer-von Mises Stat	p-Value	Test Outcome	Iterations Discarded	Half-Width	Mean	Relative Half-Width	Test Outcome
Intercept	0.0951	0.6092	Passed	0	0.0875	-0.4288	-0.2041	Failed
trt cntl	0.2343	0.2098	Passed	0	0.0630	-0.9815	-0.0641	Passed
trt drug	.	.		.	.	.	.	
Random VC(1)	0.2008	0.2659	Passed	2000	0.0655	1.8446	0.0355	Passed
Random VC(2)	0.0432	0.9160	Passed	0	0.0855	1.2928	0.0661	Passed

Raftery-Lewis Diagnostics				
Quantile=0.025 Accuracy=+/-0.005 Probability=0.95 Epsilon=0.001				
	Number of Samples			Dependence
Parameter	Burn-In	Total	Minimum	Factor
Intercept	29	32909	3746	8.7851
trt cntl	18	19114	3746	5.1025
trt drug	.	.	.	.
Random VC(1)	4	4636	3746	1.2376
Random VC(2)	4	4714	3746	1.2584

Dependence Factor
Ideally should be close to one
with all due respect to
rules of thumb
should not be > 4

Worrisome Diagnostics from Default Run, 2 of 3

Effective Sample Sizes			
Parameter	ESS	Autocorrelation Time	Efficiency
Intercept	174.1	28.7273	0.0348
trt cntl	301.6	16.5782	0.0603
trt drug	.	.	.
Random VC(1)	886.7	5.6389	0.1773
Random VC(2)	722.3	6.9225	0.1445

ESS should be >> 1000
(at least!)
Poor efficiency caused by
autocorrelation at high lags

Posterior Summaries and Intervals					
Parameter	N	Mean	Standard Deviation	95% HPD Interval	
Intercept	5000	-0.4288	0.5298	-1.4832	0.6182
trt cntl	5000	-0.9815	0.5507	-2.0139	0.1325
trt drug	0	.	.	.	.
Random VC(1)	5000	1.9221	1.2313	0.4006	4.2354
Random VC(2)	5000	**1.2928**	1.0545	0.2264	3.1124

"VC(2)" is $\hat{\sigma}^2_{ct}$
in GLIMMIX $\hat{\sigma}^2_{ct} \cong 0.02$
1.29 implausibly high

Worrisome Diagnostics from Default Run, 3 of 3

PROC BGLIMM default starting values

VS.

PROC GLIMMIX solutions

Initial Values for Fixed Effects	
Parameter	Value
Intercept	-0.3102
trt cntl	-0.4040

Solutions for Fixed Effects		
Effect	trt	Estimate
Intercept		-0.4571
trt	cntl	-0.7462

Geweke Diagnostics		
Parameter	z	Pr > |z|
Intercept	-0.4815	0.6302
trt cntl	-0.2990	0.7649
trt drug	.	.
Random VC(1)	0.5412	0.5884
Random VC(2)	0.9604	0.3369

Okay in this run
But always check

Why problems Occur

- Inadequate number of burn-in iterations
- Inadequate sampling of posterior distribution
- Inadequate thinning ($\Rightarrow$ autocorrelation)
- Inappropriate prior distributions
  - "Flat" a.k.a. "non-informative" priors often include implausible values
  - Default prior may "look where the parameter is not"

Model Information	
Burn-In Size	500
Simulation Size	5000
Thinning	1

Priors for Fixed Effects	
Parameter	Prior
Intercept	Constant
trt cntl	Constant

Priors for Scale and Covariance Parameters	
Parameter	Prior
Random Cov (Diag)	Inverse Gamma (Shape=2, Scale=2)

Default Prior

- Inverse gamma (shape=2, scale=2)
- Most likely values (e.g. obtain from GLIMMIX run)
  - clinic variance (σ_c^2) 2.0
  - clinic × treatment variance (σ_{ct}^2) <0.1
- Plot of inv-gamma(2,2)



Selection of “use what you know” priors

- Method suggested by Christiansen, et al. (2011)
- Specify the following
  - most likely value of parameter
  - extremes below or above which parameter value implausible
- Find distribution
  - whose mode ≅ most likely value
  - whose 1^{st} and 99^{th} quantiles ≅ upper and lower extremes
- Typical approach for variance
  - use precision = 1/variance
  - inv-gamma prior for variance ⟹ gamma prior for precision

Christiansen, et al. method for σ_{ct}^2

- Most likely value
  - GLIMMIX default (pseudo-likelihood) $\hat{\sigma}_{ct}^2 = 0.06$
  - GLIMMIX quadrature $\hat{\sigma}_{ct}^2 = 0.01$
  - $\Rightarrow$ most likely value for precision between 16.7 and 100
- Example program to identify candidate priors

```
data gamma_prior;
 mode=40;
 /* A=shape parameter */
 /* B=scale parameter */
 do a=1.01,1.05,1.1,1.25,1.5,2,3,5,10,20;
  b=mode/(a-1);
  do prec=0 to 200;
   var=1/prec;
   pdf_gamma=pdf("gamma",prec,a,b);
   cdf_gamma=cdf("gamma",prec,a,b);
  output;
 end;
 end;
```

```
proc sort data=gamma_prior; by a;
proc print;
 where (0.008<cdf_gamma<0.012
  or 0.988<cdf_gamma<0.992);
run;
proc gplot data=gamma_prior;
 by a;
 plot pdf_gamma*prec/href=40;
 plot cdf_gamma*prec/vref=0.01,0.99;
run;
```

Trial-and-error result

- Gamma(shape=3, scale=20)
- translates to inv-gamma(shape=3,scale=0.05) for σ_{ct}^2
- plots







Revised PROC BGLIMM

- New options in **bold**
- Increase
  - burn-in: **NBI**,
  - posterior sampling: **NMC**
  - thinning: **THIN**
- Priors
  - for TRT effects η, τ_{CNTL}
    - **LIST:** starting values
    - **COEFFPRIOR:** $\boldsymbol{N(0, VAR)}$
  - for variance of random effect: **COVPRIOR**
- Output for post-processing
  - OUTPOST
  - ODS OUTPUT

```
proc bglimm data=clinics seed=81152097
   nbi=2500 nmc=100000 thin=10
   plots=(trace autocorr density)
   diagnostics=all dic outpost=cout;
 class clinic trt;
 model fav/Nij=trt /
   init=(list=(-0.46 -0.75) pinit)
   coeffprior=normal(var=9);
 random  intercept  / subject=clinic
   covprior=igamma(shape=1.5, scale=1);
 random  trt / subject=clinic
   covprior=igamma(shape=3, scale=0.05);
 estimate "CNTL" intercept 1 trt 1 0;
 estimate "DRUG" intercept 1 trt 0 1;
 estimate "log odds-ratio" trt 1 -1;
 ods output estimates=model_scale;
 title "BGLIMM - all needed options";
run;
```

Selected Diagnostic Listing

Effective Sample Sizes			
Parameter	ESS	Autocorrelation Time	Efficiency
Intercept	2952.3	3.3872	0.2952
trt cntl	10000.0	1.0000	1.0000
trt drug	.	.	.
Random1 Var	7644.9	1.3081	0.7645
Random2 Var	8756.1	1.1421	0.8756

Raftery-Lewis Diagnostics				
Quantile=0.025 Accuracy=+/-0.005 Probability=0.95 Epsilon=0.001				
	Number of Samples			Dependence
Parameter	Burn-In	Total	Minimum	Factor
Intercept	6	6350	3746	1.6951
trt cntl	2	3834	3746	1.0235
trt drug	.	.	.	.
Random1 Var	2	3803	3746	1.0152
Random2 Var	2	3650	3746	0.9744









Posterior Summary Listing

- Variance components
  - $\hat{\sigma}_c^2 = 2.1165$ (Random1)
  - $\hat{\sigma}_{ct}^2 = 0.0261$ (Random2)
- With effects model
  - $\hat{\eta} = -0.4491$ estimates $\boldsymbol{logit(\hat{p}_{DRUG})}$
  - $\hat{\tau}_{CNTL} = -0.7454$ estimates $\boldsymbol{log(odds\ ratio)}$
- HPD
  - "highest posterior density"
  - narrowest credible interval

Posterior Summaries and Intervals

Parameter	N	Mean	Standard Deviation	95% HPD Interval	
Intercept	10000	-0.4491	0.5504	-1.6003	0.5728
trt cntl	10000	-0.7454	0.3053	-1.3201	-0.1303
trt drug	0	.	.	.	.
Random1 Var	10000	**2.1165**	1.5208	0.3744	4.6572
Random2 Var	10000	**0.0261**	0.0281	0.00444	0.0639

Results from ESTIMATE Statements

Label	Mean	Standard Deviation	95% HPD Interval	
CNTL	-1.1946	0.5612	-2.3136	-0.0881
DRUG	-0.4491	0.5504	-1.6003	0.5728
log odds-ratio	-0.7454	0.3053	-1.3201	-0.1303

analog to 95% confidence interval

Posterior percentiles

- Add **STATISTICS** option to PROC statement
- "50" (50th percentile) = median (alternative est)
- "2.5" and "97.5" percentiles give symmetric credible interval

```
proc bglimm data=clinics seed=81152097
    nbi=2500 nmc=100000 thin=10
    plots=(trace autocorr density)
    diagnostics=all
    statistics(percent=(2.5 50 97.5))
    outpost=cout;
```

Posterior Summaries

Parameter	N	Mean	Standard Deviation	Percentiles 2.5	50	97.5
Intercept	10000	-0.4491	0.5504	-1.5574	-0.4418	0.6304
trt cntl	10000	-0.7454	0.3053	-1.3539	-0.7410	-0.1619
trt drug	0	.	.	.	.	.
Random1 Var	10000	2.1165	1.5208	0.6293	1.7386	5.7644
Random2 Var	10000	0.0261	0.0281	0.00701	0.0193	0.0845

Post-processing

- PROC BGLIMM ESTIMATE statement has no ILINK option
- To obtain **data scale** estimates & intervals, e.g.
  - $\hat{p}_{CNTL}$
  - $\hat{p}_{DRUG}$
  - odds ratio
- Use post-processing data steps
- Two approaches
  - OUTPOST and %SUMINT macro
  - ODS OUTPUT and follow-up DATA step
  - OUTPOST uses posterior sample data set , ODS uses summary statistics

Post-processing 1: OUTPOST + %SUMINT

- Post-processing statements use OUTPOST=COUT

```
data datasc;
 set cout;
 pr_cntl=logistic(cntl);
 pr_drug=logistic(drug);
 ProbDiff=logistic(drug)-logistic(cntl);
 OddsRatio=exp(log_odds_ratio);
run;
%sumint(data=datasc, var=pr_cntl:
         pr_drug: ProbDiff: OddsRatio)
```

- Listing

Posterior Summaries and Intervals

Parameter	N	Mean	Standard Deviation	95% HPD Interval	
OddsRatio	10000	0.4970	0.1532	0.2251	0.8004
pr_cntl	10000	0.2461	0.0990	0.0750	0.4475
pr_drug	10000	0.3965	0.1224	0.1679	0.6394
ProbDiff	10000	0.1504	0.0667	0.0248	0.2800

Post-processing 2

- Post-processing statements using ODS OUTPUT data set

```
data datascale;
 set modelscale;
 if label='CNTL'
    or label='DRUG';
 prob=1/(1+exp(-Mean));
 lower=1/(1+exp(-HPDLower));
 upper=1/(1+exp(-HPDUpper));
```

```
data oddsratio;
 set modelscale;
 if label='log odds_ratio';
 OddsRatio=exp(Mean);
 lower=exp(HPDLower);
 upper=exp(HPDUpper);
proc print data=datascale;
proc print data=oddsratio;
run;
```

- listing

Obs	Label	Mean	StdDev	HPDLower	HPDUpper	prob	lower	upper
1	CNTL	-1.1946	0.5612	-2.3136	-0.0881	0.23245	0.09001	0.47798
2	DRUG	-0.4491	0.5504	-1.6003	0.5728	0.38957	0.16793	0.63942

Obs	Label	Mean	StdDev	HPDLower	HPDUpper	OddsRatio	lower	upper
1	log odds-ratio	-0.7454	0.3053	-1.3201	-0.1303	0.47452	0.26712	0.87780

Example 2

- Split-plot Experiment
  - Whole Plot Factor **METHOD** – 2 levels
  - Whole Plots in 6 fields
  - Split Plot Factor **MIX** – 2 levels
- Count Data
- *SAS for Mixed Models* (2018)
- Section 13.3.4

Field	W P	Treatments	
1	1	11	12
	2	21	22
2	3	11	12
	4	21	22
3	5	11	12
	6	21	22
4	7	11	12
	8	21	22
5	9	11	12
	10	21	22
6	11	11	12
	12	21	22

Re-purposed ANOVA Sources of Variation

Experiment Design		Trt Design		Combined	
Source	**d.f.**	**Source**	**d.f.**	**Source**	**d.f.**
field	5			field	5
		method	1	method	1
wp(field)	6			wp(field) | method a.k.a. field x method	6-1=5
		mix	1	mix	1
		method x mix	1	method x mix	1
sp(wp)	12	“parallels”	20	sp(wp) | mix, method a.k.a. sp error	12-2=10
Total	**23**	**Total**	**23**	**Total**	**23**

Poisson-gamma Process ⇒Negative Binomial GLMM

Combined	
Source	**d.f.**
field	5
method	1
wp(blk) | method a.k.a. field x method	6-1=5
mix	1
method x mix	1
sp(wp) | method, mix a.k.a. sp error	12-2=10
Total	**23**

Linear Predictor

$$\eta + \alpha_i + \beta_j + (\alpha\beta)_{ij} + \mathbf{b}_k + \mathbf{w}_{ik}$$

$$\mathbf{b_k} \text{ i.i.d. } N\left(0, \sigma_B^2\right)$$

$$\mathbf{w_{ik}} \text{ i.i.d. } N\left(0, \sigma_W^2\right)$$

$$y_{ijk} \mid b_k, w_{ik}, \mathbf{u_{ijk}} \sim \text{Poisson}\left(\lambda_{ijk} u_{ijk}\right)$$

$$u_{ijk} \sim Gamma\left(\phi, 1/\phi\right)$$

Log Link

$$\eta_{ijk} = \log\left(\lambda_{ijk}\right)$$

Negative Binomial GLMM

Combined	
Source	**d.f.**
field	5
method	1
wp(blk) | method a.k.a. field x method	6-1=5
mix	1
method x mix	1
sp(wp) | method, mix a.k.a. sp error	12-2=10
Total	**23**

Linear Predictor

$$\eta + \alpha_i + \beta_j + (\alpha\beta)_{ij} + \mathbf{b_k} + \mathbf{w_{ik}}$$

$$\mathbf{b_k} \text{ i.i.d. } N\left(0, \sigma_B^2\right)$$

$$\mathbf{w_{ik}} \text{ i.i.d. } N\left(0, \sigma_W^2\right)$$

$$y_{ijk} \mid b_k, w_{ik} \sim \text{Neg Bin}\left(\lambda_{ijk}, \phi\right)$$

Log Link

$$\eta_{ijk} = \log\left(\lambda_{ijk}\right)$$

Neg Bin GLMM – cell means (full rank) form

Combined	
Source	**d.f.**
field	5
method	1
wp(blk) | method a.k.a. field x method	6-1=5
mix	1
method x mix	1
sp(wp) | method, mix a.k.a. sp error	12-2=10
Total	**23**

Linear Predictor

$$\eta_{ij} + \mathbf{b_k} + \mathbf{w_{ik}}$$

$$\mathbf{b_k} \text{ i.i.d. } N\left(0, \sigma_B^2\right)$$

$$\mathbf{w_{ik}} \text{ i.i.d. } N\left(0, \sigma_W^2\right)$$

$$y_{ijk} \mid b_k, w_{ik} \sim \text{Neg Bin}\left(\lambda_{ijk}, \phi\right)$$

Log Link

$$\eta_{ijk} = \log\left(\lambda_{ijk}\right)$$

η_{ij} replaces $\eta + \alpha_i + \beta_j + \alpha\beta_{ij}$

PROC BGLIMM Statements

```
proc bglimm data=sp_counts plots=(trace autocorr density)
   seed=20210107    diagnostics=all dic
   nbi=2500 nmc=500000 thin=50 outpost=model_scale;
 class field method MeMx;
 model count=MeMx / noint distribution=negbin
   init=(list=(2.5, 2.9, 0.8, 2.0)) scaleprior=gamma(shape=20,iscale=25);
 random intercept method/ subject=field
   covprior=igamma(shape=4,scale=3);
 estimate "LSM_11" MeMx 1 0 0 0;
 estimate "LSM_12" MeMx 0 1 0 0;
 estimate "LSM_21" MeMx 0 0 1 0;
 estimate "LSM_22" MeMx 0 0 0 1;
 estimate "interaction" MeMx 1 -1 -1 1;
 estimate "Method Main Effect" MeMx 1 1 -1 -1 / divisor=2;
 estimate "Mix Main Effect" MeMx 1 -1 1 -1 / divisor=2;
 ods output estimates=link_scale;
 title "Negative Binomial GLMM";
run;
```

BGLIMM Results

- Diagnostics okay
  (not shown here)
- Listing

Posterior Summaries and Intervals					
Parameter	N	Mean	Standard Deviation	95% HPD Interval	
MeMx 11	10000	2.6048	0.7728	1.0057	4.0540
MeMx 12	10000	3.1033	0.7731	1.5136	4.5331
MeMx 21	10000	1.1871	0.7130	-0.1965	2.6263
MeMx 22	10000	2.3487	0.6749	1.0120	3.6709
Scale	10000	0.7647	0.1675	0.4520	1.0961
Random VC(1)	10000	1.5829	0.8921	0.4250	3.3169
Random VC(2)	10000	0.9997	0.5746	0.2646	2.0805

Results from ESTIMATE Statements				
Label	Mean	Standard Deviation	95% HPD Interval	
LSM_11	2.6048	0.7728	1.0057	4.0540
LSM_12	3.1033	0.7731	1.5136	4.5331
LSM_21	1.1871	0.7130	-0.1965	2.6263
LSM_22	2.3487	0.6749	1.0120	3.6709
interaction	0.6631	0.8228	-1.0406	2.2255
Method Main Effect	1.0861	0.6229	-0.1337	2.3246
Mix Main Effect	-0.8301	0.4030	-1.6100	-0.0423

Post-processing 1: with OUTPOST

```
data data_scale;
 set model_scale;
 Lambda_11=exp(MeMx_11);
 Lambda_12=exp(MeMx_12);
 Lambda_21=exp(MeMx_21);
 Lambda_22=exp(MeMx_22);
 Method_Diff_Ratio=
  exp(Method_Main_Effect);
 Mix_Diff_Ratio=
  exp(Mix_Main_Effect);

%sumint(data=data_scale,
  var=Lambda_11: Lambda_12:
  Lambda_21: Lambda_22:
  Method_Diff_Ratio:
  Mix_Diff_Ratio)
```

Posterior Summaries and Intervals

Parameter	N	Mean	Standard Deviation	95% HPD Interval	
Mix_Diff_Ratio	10000	0.4725	0.1947	0.1613	0.8636
Lambda_11	10000	18.4369	19.2826	1.1794	48.0516
Lambda_12	10000	30.4441	32.9238	2.0594	78.4440
Lambda_21	10000	4.2326	3.5203	0.2555	10.5415
Lambda_22	10000	13.2038	10.5077	1.2347	32.7550
Method_Diff_Ratio	10000	3.6258	3.0682	0.5600	8.5199

Post-processing 2: ODS OUTPUT + data steps

```
data LS_means;
 set link_scale;
 if label="LSM_11" or
  label="LSM_12" or
  label="LSM_21" or
  label="LSM_22";
 Lambda_Est=exp(Mean);
 Lambda_Lower=exp(HPDLower);
 Lambda_Upper=exp(HPDUpper);
 title "Method x Mix LSMeans";
proc print data=LS_means;
run;
data diff_ratios;
 set link_scale;
 if label="Method Main Effect" or
  label="Mix Main Effect";
 Ratio_Est=exp(Mean);
 Ratio_Lower=exp(HPDLower);
 Ratio_Upper=exp(HPDUpper);
 title "Diff Ratios";
proc print data=diff_ratios;
run;
```

Obs	Label	Lambda_Est	Lambda_Lower	Lambda_Upper
1	LSM_11	13.5282	2.73376	57.6280
2	LSM_12	22.2715	4.54289	93.0480
3	LSM_21	3.2776	0.82158	13.8230
4	LSM_22	10.4720	2.75115	39.2874

Obs	Label	Ratio_Est	Ratio_Lower	Ratio_Upper
1	Method Main Effect	2.96279	0.87488	10.2222
2	Mix Main Effect	0.43603	0.19990	0.9586

Example 3

- Repeated measures (a.k.a. longitudinal) data
- Example 8.2 in *SAS for Mixed Models* (2018)
  - data from Littell, Pendergast and Natarajan (2000)
  - also appears in *SAS for Linear Models* (2002) and PROC BGLIMM documentation
- Three treatments (labeled DRUG in data set)
  - two experimental nasal treatments (A and C)
  - one placebo (P)
- Response variable
  - FEV1 – a measure of breathing comfort
  - Baseline measurement (BASEFEV) and at 0,1,2,...8 hours after treatment
- 24 patients assigned to each DRUG

Re-purposed ANOVA Sources of Variation

Experiment Design		Trt Design		Combined	
Source	**d.f.**	**Source**	**d.f.**	**Source**	**d.f.**
		drug	3-1=2	drug	2
patient	72-1=71			**patient| drug aka between**	**71-2=69**
		hour	8-1=7	hour	7
		drug x hour	14	drug x hour	14
occasion(patient)	72(8-1) = 504			**occ(pt) | drug aka within**	**504-7-14= 483**
Total	**576-1=575**	**Total**	**575**	**Total**	**575**

Linear Predictor: $\eta_{ijk} = \alpha_i + \mathbf{s_{ij}} + \tau_k + (\alpha\tau)_{ik}$

Within: $\begin{bmatrix} w_{ij1} & w_{ij2} & w_{ij3} & w_{ij4} & w_{ij5} & w_{ij6} \end{bmatrix}$ correlated

Interaction Plot, Objectives, PROC BGLIMM Statements

- FEV1: higher value ⟹ greater breathing comfort
- Objectives
  - do experimental drugs (A and C) improve breathing comfort relative to placebo (P)?
  - how long does improvement last?
  - objectives addressed by drug x hour effect
- BGLIMM Documentation
  - “model 2” fev1=basefev drug hour;
  - “model 3” fev1=basefev drug | hour;
  - documentation: “minor difference”
  - NOT minor: objective *requires* model 3!!



Alternative Models

- Generic Model
  - $y|between\ subject \sim N(\mu, R)$
  - between subject [patient(drug)]: $s_{ij} \sim N(0, \sigma_s^2)$
  - within subject covariance $R = I_{72} \otimes R_{ij}$
  - R_{ij} is 8×8 covariance matrix, e.g. AR(1)
- Effects model for μ
  - $\eta + \beta X_{ij} + \alpha_i + \tau_k + \alpha\tau_{ik} + s_{ij}$; α=drug; τ=hour; X_{ij}=baseline FEV1
  - cell means version: $\eta + \beta X_{ij} + \alpha\tau_{ik} + s_{ij}$
- Unequal slopes regression
  - $\eta + \beta X_{ij} + \alpha_i + \beta_i H + s_{ij}$; H=hour

PROC BGLIMM Statements – Effects Model

Focus on Simple Effect of Drug at each Hour

```
proc bglimm data=fev1uni nmc=10000 seed=44672057
            plots=(trace autocorr density) diagnostics=all;
   class Drug Patient Hour;
   model FEV1 = BaseFev1 Drug|Hour;
   random int / subject=Patient;
   repeated Hour / subject=Patient(Drug) type=ar(1) r rcorr;
   estimate "A vs P at hour 1" Drug 1 0 -1 Drug*Hour 1 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0  -1 0 0 0 0 0 0 0;
   estimate "C vs P at hour 1" Drug 0 1 -1 Drug*Hour 0 0 0 0 0 0 0 0  1 0 0 0 0 0 0 0  -1 0 0 0 0 0 0 0;
   estimate "A vs P at hour 2" Drug 1 0 -1 Drug*Hour 0 1 0 0 0 0 0 0  0 0 0 0 0 0 0 0  0 -1 0 0 0 0 0 0;
   estimate "C vs P at hour 2" Drug 0 1 -1 Drug*Hour 0 0 0 0 0 0 0 0  0 1 0 0 0 0 0 0  0 -1 0 0 0 0 0 0;
   estimate "A vs P at hour 3" Drug 1 0 -1 Drug*Hour 0 0 1 0 0 0 0 0  0 0 0 0 0 0 0 0  0 0 -1 0 0 0 0 0;
   estimate "C vs P at hour 3" Drug 0 1 -1 Drug*Hour 0 0 0 0 0 0 0 0  0 0 1 0 0 0 0 0  0 0 -1 0 0 0 0 0;
   estimate "A vs P at hour 4" Drug 1 0 -1 Drug*Hour 0 0 0 1 0 0 0 0  0 0 0 0 0 0 0 0  0 0 0 -1 0 0 0 0;
   estimate "C vs P at hour 4" Drug 0 1 -1 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 1 0 0 0 0  0 0 0 -1 0 0 0 0;
   estimate "A vs P at hour 5" Drug 1 0 -1 Drug*Hour 0 0 0 0 1 0 0 0  0 0 0 0 0 0 0 0  0 0 0 0 -1 0 0 0;
   estimate "C vs P at hour 5" Drug 0 1 -1 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 1 0 0 0  0 0 0 0 -1 0 0 0;
   estimate "A vs P at hour 6" Drug 1 0 -1 Drug*Hour 0 0 0 0 0 1 0 0  0 0 0 0 0 0 0 0  0 0 0 0 0 -1 0 0;
   estimate "C vs P at hour 6" Drug 0 1 -1 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 1 0 0  0 0 0 0 0 -1 0 0;
   estimate "A vs P at hour 7" Drug 1 0 -1 Drug*Hour 0 0 0 0 0 0 1 0  0 0 0 0 0 0 0 0  0 0 0 0 0 0 -1 0;
   estimate "C vs P at hour 7" Drug 0 1 -1 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 0 1 0  0 0 0 0 0 0 -1 0;
   estimate "A vs P at hour 8" Drug 1 0 -1 Drug*Hour 0 0 0 0 0 0 0 1  0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 -1;
   estimate "C vs P at hour 8" Drug 0 1 -1 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 1  0 0 0 0 0 0 0 -1;
run;
```

Alternative Effects Model – Cell Means Form

obtain drug x hour LSMeans – coefficients from GLIMMIX LSMEANS E option

```
proc bglimm data=fev1uni nmc=10000 seed=44672057 plots=(trace autocorr density) diagnostics=all;
   class Drug Patient Hour;
   model FEV1 = BaseFev1 Drug*Hour;
   random int / subject=Patient;
   repeated Hour / subject=Patient(Drug) type=ar(1) r rcorr;
   estimate "LSM A at hour 1"  intercept 1 basefev1 2.6493 Drug*Hour 1 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM C at hour 1"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  1 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM P at hour 1"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0  1 0 0 0 0 0 0 0;
   estimate "LSM A at hour 2"  intercept 1 basefev1 2.6493 Drug*Hour 0 1 0 0 0 0 0 0  0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM C at hour 2"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 1 0 0 0 0 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM P at hour 2"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0  0 1 0 0 0 0 0 0;
   estimate "LSM A at hour 3"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 1 0 0 0 0 0  0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM C at hour 3"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 1 0 0 0 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM P at hour 3"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0  0 0 1 0 0 0 0 0;
   estimate "LSM A at hour 4"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 1 0 0 0 0  0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM C at hour 4"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 1 0 0 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM P at hour 4"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0  0 0 0 1 0 0 0 0;
   estimate "LSM A at hour 5"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 1 0 0 0  0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM C at hour 5"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 1 0 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM P at hour 5"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0  0 0 0 0 1 0 0 0;
   estimate "LSM A at hour 6"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 1 0 0  0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM C at hour 6"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 1 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM P at hour 6"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0  0 0 0 0 0 1 0 0;
   estimate "LSM A at hour 7"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 1 0  0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM C at hour 7"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 0 1 0  0 0 0 0 0 0 0 0;
   estimate "LSM P at hour 7"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0  0 0 0 0 0 0 1 0;
   estimate "LSM A at hour 8"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 1  0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0;
   estimate "LSM C at hour 8"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 1  0 0 0 0 0 0 0 0;
   estimate "LSM P at hour 8"  intercept 1 basefev1 2.6493 Drug*Hour 0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 1;
```

PROC BGLIMM Statements – Regression Model

Unequal Slope over Hour by Drug – Test Regression Coefficients

```
data fev1uni; set fev1uni;
   H=hour;
proc bglimm data=fev1uni nmc=10000 seed=44672057
            plots=(trace autocorr density) diagnostics=all;
   class Drug Patient Hour;
   model FEV1 = BaseFev1 drug H(drug);
   random int / subject=Patient;
   repeated Hour / subject=Patient(Drug) type=ar(1) r rcorr;
   estimate "intercept - Drug A" intercept 1 drug 1 0 0;
   estimate "intercept - Drug C" intercept 1 drug 0 1 0;
   estimate "intercept - Drug P" intercept 1 drug 0 0 1;
   estimate "slope - Drug A" H(drug) 1 0 0;
   estimate "slope - Drug C" H(drug) 0 1 0;
   estimate "slope - Drug P" H(drug) 0 0 1;
   estimate "intercept A vs C" drug 1 -1 0;
   estimate "intercept A vs P" drug 1 0 -1;
   estimate "intercept C vs P" drug 0 1 -1;
   estimate "slope A vs C" H(drug) 1 -1 0;
   estimate "slope A vs P" H(drug) 1 0 -1;
   estimate "slope C vs P" H(drug) 0 1 -1;
run;
```

Results

Effects Model

LSMeans

Results from ESTIMATE Statements				
Label	Mean	Standard Deviation	95% HPD Interval	
LSM A at hour 1	3.4735	0.1400	3.1911	3.7370
LSM C at hour 1	3.6820	0.1387	3.4068	3.9512
LSM P at hour 1	2.8178	0.1395	2.5410	3.0889
LSM A at hour 2	3.3965	0.1405	3.1281	3.6790
LSM C at hour 2	3.6183	0.1382	3.3483	3.8947
LSM P at hour 2	2.8828	0.1397	2.6099	3.1591
LSM A at hour 3	3.1841	0.1405	2.9071	3.4586
LSM C at hour 3	3.5692	0.1394	3.2962	3.8484
LSM P at hour 3	2.8891	0.1401	2.6033	3.1555
LSM A at hour 4	3.0464	0.1411	2.7666	3.3261
LSM C at hour 4	3.4350	0.1396	3.1570	3.7070
LSM P at hour 4	2.8619	0.1402	2.5855	3.1373
LSM A at hour 5	3.0533	0.1415	2.7816	3.3398
LSM C at hour 5	3.2415	0.1386	2.9692	3.5131
LSM P at hour 5	2.7595	0.1406	2.4729	3.0256
LSM A at hour 6	2.9797	0.1411	2.6956	3.2489
LSM C at hour 6	3.0773	0.1386	2.8061	3.3493
LSM P at hour 6	2.8064	0.1416	2.5306	3.0907
LSM A at hour 7	2.8691	0.1403	2.5996	3.1529
LSM C at hour 7	2.9692	0.1385	2.6919	3.2316
LSM P at hour 7	2.7755	0.1412	2.4912	3.0458
LSM A at hour 8	2.8581	0.1395	2.5712	3.1175
LSM C at hour 8	3.0026	0.1384	2.7331	3.2743
LSM P at hour 8	2.7227	0.1403	2.4497	2.9961

Simple Effect Diffs

Results from ESTIMATE Statements				
Label	Mean	Standard Deviation	95% HPD Interval	
A vs P at hour 1	0.6557	0.0981	0.4626	0.8460
C vs P at hour 1	0.8642	0.0973	0.6796	1.0579
A vs P at hour 2	0.5137	0.0969	0.3281	0.7036
C vs P at hour 2	0.7355	0.0966	0.5482	0.9279
A vs P at hour 3	0.2950	0.0962	0.1058	0.4832
C vs P at hour 3	0.6801	0.0968	0.4904	0.8731
A vs P at hour 4	0.1845	0.0968	-0.00386	0.3702
C vs P at hour 4	0.5731	0.0969	0.3911	0.7720
A vs P at hour 5	0.2937	0.0966	0.1056	0.4811
C vs P at hour 5	0.4819	0.0968	0.2888	0.6659
A vs P at hour 6	0.1733	0.0973	-0.00990	0.3699
C vs P at hour 6	0.2709	0.0973	0.0882	0.4660
A vs P at hour 7	0.0935	0.0966	-0.0891	0.2908
C vs P at hour 7	0.1937	0.0974	0.00599	0.3858
A vs P at hour 8	0.1354	0.0971	-0.0530	0.3226
C vs P at hour 8	0.2799	0.0970	0.0866	0.4702

Results

Regression Model

Posterior Summaries and Intervals

Parameter	N	Mean	Standard Deviation	95% HPD Interval	
Intercept	10000	1.4196	0.4008	0.6234	2.1897
basefev1	10000	0.5460	0.1419	0.2685	0.8240
drug a	10000	0.6577	0.1051	0.4435	0.8524
drug c	10000	0.9276	0.1046	0.7222	1.1296
drug p	0	.	.	.	.
H*drug a	10000	-0.0886	0.0135	-0.1148	-0.0619
H*drug c	10000	-0.1034	0.0133	-0.1290	-0.0768
H*drug p	10000	-0.0151	0.0133	-0.0424	0.00932
Residual Var	10000	0.1142	0.0119	0.0924	0.1384
Residual AR(1)	10000	0.6550	0.0356	0.5863	0.7274
Random Var	10000	0.3506	0.1101	0.1787	0.5748

Results from ESTIMATE Statements

Label	Mean	Standard Deviation	95% HPD Interval	
intercept - Drug A	2.0774	0.4056	1.2770	2.8641
intercept - Drug C	2.3473	0.4021	1.5223	3.0958
intercept - Drug P	1.4196	0.4008	0.6234	2.1897
slope - Drug A	-0.0886	0.0135	-0.1148	-0.0619
slope - Drug C	-0.1034	0.0133	-0.1290	-0.0768
slope - Drug P	-0.0151	0.0133	-0.0424	0.00932
intercept A vs C	-0.2699	0.1071	-0.4790	-0.0588
intercept A vs P	0.6577	0.1051	0.4435	0.8524
intercept C vs P	0.9276	0.1046	0.7222	1.1296
slope A vs C	0.0147	0.0189	-0.0235	0.0503
slope A vs P	-0.0735	0.0189	-0.1095	-0.0356
slope C vs P	-0.0883	0.0187	-0.1272	-0.0536

Final Thoughts

PROC GLIMMIX ↔PROC BGLIMM↔PROC MCMC

- PROC BGLIMM can implement same models as GLIMMIX
  - more diagnostics (trace plots, ESS. dependence factor, etc.)
  - NBI, NMC, THIN and prior options usually need attention
  - OUTPOST & ODS OUTPUT instead of ILINK
- PROC MCMC
  - offers more flexible specification of prior distributions
  - e.g. prior centered at fixed effect starting value
  - can fit non-linear and semi-parametric mixed models
  - allows non-Gaussian random effects
  - e.g. Beta-binomial, Poisson-gamma mixed models

Thank you!

Contact Information
wstroup@unl.edu



SAS® GLOBAL FORUM 2021

#SASGF